

HS12864-15B 中文图形液晶模块说明书

感谢您关注和使用我们的液晶产品。如果您在使用中有任何疑问，请拨打我们的客户服务热线 **0755-86114312** 寻求技术支持和获取相关资料，我们竭诚为您服务。您可以登录我们的网站了解最新产品信息。或者您可以在我公司网站的留言簿栏目留下您宝贵的意见。

深圳汉昇实业有限公司

SHENZHEN HANSHENG INDUSTRIAL CO.,LTD

地 址：深圳市南山区西丽阳光社区米坑新锋工业园 2 栋 5 楼
邮 编：518055
公司主页：www.hsicm.com
电 话：0755-86114312
传 真：0755-27657419

第一章 HS12864-15B 中文图形液晶显示模块的主要特性

HS12864-15B 中文图形液晶模块的特性主要由其控制器 ST7920 决定。ST7920 同时作为控制器和驱动器，它可提供 33 路 com 输出和 64 路 seg 输出。在驱动器 ST7921 的配合下，最多可以驱动 256×32 点阵液晶。

HS12864-15 系列产品硬件特性如下：

- 提供 8 位，4 位并行接口及串行接口可选
- 并行接口适配 M6800 时序
- 自动电源启动复位功能
- 内部自建振荡源
- 64×16 位字符显示 RAM (DDRAM 最多 16 字符×4 行，LCD 显示范围 16×2 行)
- 2M 位中文字型 ROM (CGROM)，总共提供 8192 个中文字型 (16×16 点阵)
- 16K 位半宽字型 ROM (HCGROM)，总共提供 126 个西文字型 (16×8 点阵)
- 64×16 位字符产生 RAM (CGRAM)

HS12864-15 系列产品软件特性如下：

- 文字与图形混合显示功能
- 画面清除功能
- 光标归位功能
- 显示开/关功能
- 光标显示/隐藏功能
- 显示字体闪烁功能
- 光标移位功能功能
- 显示移位功能
- 垂直画面旋转功能
- 反白显示功能
- 休眠模式

中文字库选择：


ST7920-0A 内建 BIG-5 码繁体中文字型库

ST7920-0B 内建 GB 码简体中文字型库

用户在选用之前务必注明。

第二章 模块的硬件说明

1. HS12864-15B 外形图




2. HS12864-15B 外形参数表

项 目	标 准 尺 寸	单 位
模块体积	$93.0 \times 70.0 \times 13.0$	mm
定位尺寸	88.0×65.0	mm
视域尺寸	72.0×40.0	mm
字符点阵	128×64	位
点 距 离	0.52×0.52	mm
点 大 小	0.48×0.48	mm

3. HS12864-15B 接口定义

	名称	型态	电平	功能描述	
				并口	串口
2	VEE		-	液晶驱动电压输出端（可空接）	
1	VSS		-	电源地	
2	VDD		-	模块电源输入（未注明为 5V）	
3	V0/NC		-	对比度调节端，一般已设置好，可悬空	
4	RS (CS)	I	H/L	寄存器选择端： H 数据； L 指令	片选，高有效
5	R/W (SID)	I	H/L	读/写选择端： H 读； L 写	串行数据线
6	E (SCK)	I	H/L	使能信号	串行时钟输入
7	DB0	I/O	H/L	数据总线低四位	空接
8	DB1				
9	DB2				
10	DB3				
11	DB4	I/O	H/L	数据总线高四位，4 位并口 时空接	空接
12	DB5				
13	DB6				
14	DB7				
15	PSB	I	H/L	并口/串口选择：H 并口； L 串口**	
16	NC		-		
17	/RST	I	H/L	复位信号，低有效	
18	VOUT		-	液晶驱动电压输出端（可空接）	
19	LEDA	I	-	背光正	
20	LEDK	I	-	背光负	

4. 原理简图



5. 电气特性: (测试条件 $T_a=25, V_{dd}=5.0\pm 10\%$)

- 1) 输入高电平 (V_{ih}): $0.7V_{dd}\sim V_{dd}$
- 2) 输入低电平 (V_{il}): $0.6V$ max
- 3) 输出高电平 (V_{oh}): $0.8V_{dd}\sim V_{dd}$
- 4) 输出低电平 (V_{ol}): $0.4V$ max
- 5) 模块工作电流: $3\sim 5mA$ (不含背光)
- 6) 白背光工作电流: $60mA$ max
- 7) 黄绿背光工作电流: $360mA$ max


6. LCD 驱动电压及对比度调节电路

通过售后反馈, 关于对比度的调节方式, 用户的疑问比较多, 本节介绍 HS12864-15B 的对比度调节方式:

使用 ST7920 控制驱动器的液晶模块(包括 HS12864-15B), 可以使用 ST7920 内带倍压单元, 外挂 2 个极性电容, 生成 2 倍于 VD_2 的电压 (VD_2 来源于电源供电 VCC, 通常经过电阻分压, 因此 $VD_2 \leq VCC$), 通过接口的 V_{out} 脚引出。

5.0V 供电情况下, HS12864-15B 液晶模块一般不配置倍压电路, 对比度已经锁定或者由液晶上的电位器来调节 (出厂前已经校正好, 一般不需调整)。V0 和 VOUT 都悬空。

3.3V 供电情况下, HS12864-15B 液晶模块配置倍压电路, 对比度已经锁定或者由液晶上的电位器来调节 (出厂前已经校正好, 一般不需调整)。V0 和 VOUT 都悬空。用户指定需要外调对比度时, 接线方法如下图所示:



7. 串并口选择

用户通过接口的 **PSB** 脚接高或接低来选择模块是使用并口还是串口。

HS12864-15B 液晶模块上有跳点方式选择将 **PSB** 接高或接低，用户可以不再处理接口的 **PSB** 脚（用户事先需声明是用并口还是串口）。

如果模块上已经对 **PSB** 作了处理，用户再在接口上作出相反的选择，可能会造成短路，对模块造成不可修复的损坏。比如，模块上已经选择了并口，此时 **PSB** 与 **VCC** 相通，如果用户将接口的 **PSB** 与 **GND** 相连，则会烧坏线路板。

如果没有客户的特殊要求，在出厂时，我们默认不对模块上的 **PSB** 做处理，由用户自己通过接口的 **PSB** 脚来选择模块的工作模式。

8. 背光极性说明

为适应不同客户的要求，**HS12864-15B** 液晶模块的背光极性可选，既可以 19 脚正，20 脚负；也可以 19 脚负，20 脚正，由模块背后的两个 0 欧电阻的焊接方式决定极性。在出厂时，我们按照客户要求配置背光极性。如无说明，我们默认 19 脚为背光负，20 脚为背光正。

第三章ST7920 内置硬件说明

1. 中文字型产生 ROM (CGROM) 及半宽字型 ROM (HCGROM)

ST7920 的字型产生 ROM 通过 8192 个 16×16 点阵的中文字型, 以及 126 个 16×8 点阵的西文字符, 它用 2 个字节来提供编码选择, 将要显示的字符的编码写到 DDRAM 上, 硬件将依照编码自动从 CGROM 中选择将要显示的字型显示再屏幕上。

2. 字型发生 RAM (CGRAM)

ST7920 的字型产生 RAM 提供用户自定义字符生成 (造字) 功能, 可提供 4 组 16×16 点阵的空间, 用户可以将 CGROM 中没有的字符定义到 CGRAM 中。

3. 显示 RAM (DDRAM)


显示 RAM 提供 64×2 字节的空间, 最多可以控制 4 行 16 字的中文字型显示。当写入显示资料 RAM 时, 可以分别显示 CGROM, HCGROM 及 CGRAM 的字型。

三种字型的选择:

- 1) 显示半宽字型 将一个字节的编码写入 DDRAM 中, 范围是 02~7FH
- 2) 显示 CGRAM 字型 将 2 个字节的编码写入 DDRAM 中, 共有 0000H, 0002H, 0004H 及 0006H 四种编码
- 3) 显示中文字型 将 2 字节的编码写入 DDRAM 中, 先写高 8 位, 后写低 8 位范围是 A140H~D75FH (BIG5), A1A0H~F7FFH (GB)

4. 绘图 RAM

提供 64×32 个字节的存储空间（由扩充指令设定绘图 RAM 地址），最多可以控制 256×64 点阵的二维绘图缓冲空间，在更改绘图 RAM 是，由扩充指令设置 GDRAM 地址先垂直地址后水平地址（连续 2 个字节的的数据来定义垂直和水平地址），再 2 个字节的的数据给绘图 RAM（先高 8 位后低 8 位）。




5. DDRAM 内容，CGRAM 地址以及 CGRAM 内容的对照关系


DDRAM 資料 (字元代碼)				CGRAM 位址				CGRAM 資料 (高位元組)				CGRAM 資料 (低位元組)																					
B15~B4	B3	B2	B1	B0	B5	B4	B3	B2	B1	B0	D15	D14	D13	D12	D11	D10	D9	D8	D7	D6	D5	D4	D3	D2	D1	D0							
0	X	00	X	00	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	0	0	0	0	0	0						
					0	0	0	1	1	1	1	1	1	1	1	0	0	1	0	0	0	0	0	1	0	0	0	0	0	0	0		
					0	0	1	0	0	0	0	1	0	0	0	0	0	0	0	0	0	0	0	1	0	0	0	1	0	0	0	0	
					0	0	1	1	0	0	0	1	0	0	0	0	0	0	0	0	0	0	0	1	1	1	1	1	1	1	0	0	
					0	1	0	0	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	0	0	0	1	0	0	0	
					0	1	0	1	0	0	1	1	1	1	0	0	1	0	0	1	0	0	1	0	0	0	0	0	1	0	0	0	
					0	1	1	0	0	1	1	0	0	1	0	1	0	1	0	1	0	1	0	0	1	0	0	1	0	0	0	0	
					0	1	1	1	1	0	1	0	0	1	1	0	0	1	1	0	0	1	0	0	1	0	0	1	0	0	0	0	
					1	0	0	0	0	0	1	0	0	1	0	0	1	0	0	0	0	0	1	0	0	0	1	0	0	0	0	0	
					1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	0	0	0	0	0	1	0	0	1	0	0	0	0	
					1	0	1	0	0	0	1	0	0	1	0	0	1	0	0	0	0	0	0	0	1	0	0	1	0	0	0	0	
					1	0	1	1	0	0	1	1	1	1	1	0	0	0	0	0	0	0	0	0	1	0	0	1	0	0	0	0	
					1	1	0	0	0	0	1	0	0	0	0	0	0	0	0	0	0	0	0	1	0	0	0	0	0	0	0	0	
					1	1	0	1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	0	0	0	0	0	0	0	0	
					1	1	1	1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	
					0	X	01	X	01	0	0	0	0	0	0	0	0	0	0	1	1	0	0	0	0	0	0	0	0	0	1	1	0
0	0	0	1	0						0	0	1	1	0	1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	0	0	
0	0	1	0	0						0	1	0	0	0	0	0	0	0	1	0	0	1	1	0	1	0	0	1	0	0	1	0	0
0	0	1	1	0						1	0	1	1	1	0	1	1	0	1	1	0	1	0	0	1	0	0	1	0	0	1	0	0
0	1	0	0	1						0	0	0	0	0	0	0	0	0	0	1	0	1	0	0	1	0	0	1	0	0	1	0	0
0	1	0	1	0						1	1	1	1	1	1	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0
0	1	1	0	0						1	0	0	0	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0
0	1	1	1	0						1	1	1	1	1	1	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0
1	0	0	0	0						1	0	0	0	0	0	0	0	0	0	1	0	0	1	0	0	1	0	0	1	0	0	0	0
1	0	0	1	0						1	1	1	1	1	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	1	0	0	0
1	0	1	0	0						1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	0	0	1	0	0	0	0	0
1	0	1	1	0						1	1	1	1	1	1	1	0	0	0	0	0	0	0	0	1	0	0	1	0	0	0	0	0
1	1	0	0	1						0	1	0	0	0	0	0	0	0	0	1	0	1	0	0	1	0	0	1	0	0	0	0	0
1	1	0	1	0						1	0	0	0	0	0	0	0	0	0	1	0	0	0	0	1	0	0	0	0	0	0	0	0
1	1	1	1	0						0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

第四章 时序


1. 8 位并口写操作时序图




2. 8 位并口读操作时序图




3. 4 位并口时序图图



4. 串口时序图



5. 外部复位时序图



XRESET pulse width	T_{rw}	10us
RESET start time	T_{res}	50ns

第五章 指令说明

1. 指令表 1 (RE=0, 基本指令集)

指令名称	控制信号		控制代码								执行时间
	RS	R/W	D7	D6	D5	D4	D3	D2	D1	D0	
清除显示	0	0	0	0	0	0	0	0	0	1	1.6 ms
地址归 0	0	0	0	0	0	0	0	0	1	X	72us
进入设定点	0	0	0	0	0	0	0	1	I/D	S	72us
显示开关设置	0	0	0	0	0	0	1	D	C	B	72us
移位控制	0	0	0	0	0	1	S/C	R/L	X	X	72us
功能设定	0	0	0	0	1	DL	X	0/RE	X	X	72us
设定 CGRAM 地址	0	0	0	1	A5	A4	A3	A2	A1	A0	72us
设定 DDRAM 地址	0	0	1	0	A5	A4	A3	A2	A1	A0	72us

读忙标志和地址	0	1	BF	A6	A5	A4	A3	A2	A1	A0	72 μ s
写显示数据	1	0	显示数据								72 μ s
读显示数据	1	1	显示数据								72 μ s

2. 指令表 2 (RE=1, 扩充指令集)

指令名称	控制信号		控制代码								执行时间
	RS	R/W	D7	D6	D5	D4	D3	D2	D1	D0	
待命模式	0	0	0	0	0	0	0	0	0	1	72 μ s
滚动地址或 RAM 地址选择	0	0	0	0	0	0	0	0	1	SR	72 μ s
反白显示	0	0	0	0	0	0	0	1	R1	R0	72 μ s
睡眠模式	0	0	0	0	0	0	1	SL	X	X	72 μ s
扩充功能设定	0	0	0	0	1	DL	X	1/RE	G	0	72 μ s
设定绘图 RAM 地址	0	0	1	0	0	0	A3	A2	A1	A0	72 μ s
A6				A5	A4	A3	A2	A1	A0		

备注：当 ST7920 在接受指令前，MCU 必须先确认 ST7920 处于非忙状态。即读取 BF = 0，才能接受新的指令；如果在送出一条指令前不检查 BF 状态，则需要延时一段时间，以确保上一条指令执行完毕，具体指令执行时间参照指令表。

“RE”是基本指令集与扩充指令集的选择控制位，当变更“RE”的状态后，以后的指令维持在最后的状态。除非再次变更“RE”的状态，否则使用相同的指令集时，不需要重新设置“RE”。

3. 基本指令详细说明表

1) 清除显示 (CLEAR)

格式

0	0	0	0	0	0	0	0	1
---	---	---	---	---	---	---	---	---

将 DDRAM 填满“20H”（空格）代码，并且设定 DDRAM 的地址计数器 (AC) 为 00H；更新设置进入设定点将 I/D 设为 1，游标右移 AC 加 1。

2) 地址归 0 (HOME)

格式

0	0	0	0	0	0	0	1	X
---	---	---	---	---	---	---	---	---

设定 DDRAM 的地址寄存器为 00H，并且将游标移到开头原点位置；这个指令并不改变 DDRAM 的内容。

3) 进入设定点 (ENTRY MODE SET) 初始值：06H

格式

0	0	0	0	0	1	I/D	S
---	---	---	---	---	---	-----	---

指定在显示数据的读取与写入时，设定游标的移动方向及指定显示的移位

I/D=1，游标右移，DDRAM 地址计数器 (AC) 加 1

I/D=0，游标左移，DDRAM 地址计数器 (AC) 减 1

S: 显示画面整体位移

S	I/D	功能描述
H	H	画面整体左移
H	L	画面整体右移

4) 显示开关设置 (DISPLAY STATUS) 初始值: 08H

格式	0	0	0	0	1	D	C	B
----	---	---	---	---	---	---	---	---

控制整体显示开关, 游标开关, 游标位置显示反白开关

D=1, 整体显示开; D=0, 整体显示关, 但是不改变 DDRAM 内容

C=1, 游标显示开; C=0, 游标显示关

B=1, 游标位置显示反白开, 将游标所在地址上的内容反白显示; B=0, 正常显示

5) 游标或显示移位控制(CURSORM AND DISPLAY SHIFT CONTROL)

初始值: 0001 XXXX B (X=0,1)

格式	0	0	0	1	S/C	R/L	X	X
----	---	---	---	---	-----	-----	---	---

这条指令不改变 DDRAM 的内容

S/C	R/L	方向	AC 的值
L	L	游标向左移动	AC=AC-1
L	H	游标向右移动	AC=AC+1
H	L	显示向左移动, 游标跟着移动	AC=AC
H	H	显示向右移动, 游标跟着移动	AC=AC

6) 功能设定(FUNCTION SET) 初始值: 0011 X0XX B (X=0,1)

格式	0	0	1	DL	X	0/RE	X	X
----	---	---	---	----	---	------	---	---

DL: 8/4 位接口控制位

DL=1, 8 位 MPU 接口; DL=0, 4 位 MPU 接口

RE: 指令集选择控制位

RE=1, 扩充指令集; RE=0, 基本指令集

同一指令的动作不能同时改变 DL 和 RE, 需先改变 DL 再改变 RE 才能确保设置正确

7) 设定 CGRAM 地址

格式	0	1	A5	A4	A3	A2	A1	A0
----	---	---	----	----	----	----	----	----

设定 CGRAM 地址到地址计数器 (AC), AC 范围为 00H~3FH 需确认扩充指令中 SR=0 (滚动位置或 RAM 地址选择)

8) 设定 DDRAM 地址

格式	1	0	A5	A4	A3	A2	A1	A0
----	---	---	----	----	----	----	----	----

设定 DDRAM 地址到地址计数器 (AC)

第一行 AC 范围 80H~8FH

第二行 AC 范围 90H~9FH

备注: ST7920 控制器的 128×64 点阵液晶其实原理上等同 256×32 点阵, 第

三行对应的 DDRAM 地址紧接第一行；第四行对应的 DDRAM 地址紧接第二行。

用户在使用行反白功能时，如果第一行反白，第三行必然反白。第二行反白，第四行必然反白。这是正常现象。

9) 读取忙标志和地址(RS=0,R/W=1)

格式	BF	A6	A5	A4	A3	A2	A1	A0
----	----	----	----	----	----	----	----	----

读取忙标志以确定内部动作是否完成，同时可以读出地址计数器（AC）的值

10) 写显示数据到 RAM(RS=1,R/W=0)

格式	D7	D6	D5	D4	D3	D2	D1	D0
----	----	----	----	----	----	----	----	----

当显示数据写入后会使得 AC 改变，每个 RAM（CGRAM, DDRAM）地址都可以连续写入 2 个字节的显示数据，当写入第二个字节时，地址计数器（AC）的值自动加一。

11) 读取显示 RAM 数据（RS=1, R/W=1）

格式	D7	D6	D5	D4	D3	D2	D1	D0
----	----	----	----	----	----	----	----	----

读取后会使得 AC 改变

设定 RAM（CGRAM, DDRAM）地址后，先要 Dummy read 一次后才能读取到正确的显示数据，第二次读取不需要 Dummy read，除非重新设置了 RAM 地址

4. 扩充指令详细说明表

1) 待命模式

格式	0	0	0	0	0	0	0	1
----	---	---	---	---	---	---	---	---

进入待命模式，执行如何其它指令都可以结束待命模式；该指令不能改变 RAM 的内容。

2) 卷动位置或者 RAM 地址选择 初始值：02H

格式	0	0	0	0	0	0	1	SR
----	---	---	---	---	---	---	---	----

当 SR=1 时，允许输入垂直卷动地址

当 SR=0 时，允许设定 CGRAM 地址（基本指令）

3) 反白显示 初始值：04H

格式	0	0	0	0	0	1	0	R0
----	---	---	---	---	---	---	---	----

选择 2 行中的任意一行作反白显示，并可决定反白与否。R0 初始值为 0，第一次执行时为反白显示，再次执行时为正常显示

通过 R0 选择要作反白处理的行：

R0=0 第一行， R0=1 第二行

说明：参考基本指令详细说明中的 DDRAM 地址说明

128×64 点阵的液晶执行反白功能时实用意义不大，因为一三行连在一起，二四行连在一起，用户对第一行执行反白显示操作时，第三行必然也反白显示。

4) 睡眠模式

初始值: 0000 10XXB(X=0,1)

格式	0	0	0	0	1	SL	0	0
----	---	---	---	---	---	----	---	---

SL=1, 脱离睡眠模式

SL=0, 进入睡眠模式

5) 扩充功能设定

初始值: 001 DL X100 B (DL=1,8BIT 并口;DL=0,4BIT 并口 X=0,1)

格式	0	0	1	DL	X	RE	G	X
----	---	---	---	----	---	----	---	---

DL: 8/4 位接口控制位

DL=1, 8 位 MPU 接口; DL=0, 4 位 MPU 接口

RE: 指令集选择控制位

RE=1, 扩充指令集; RE=0, 基本指令集

G: 绘图显示控制位

G=1, 绘图显示开; G=0, 绘图显示关

同一指令的动作不能同时改变 RE 及 DL、G, 需先改变 DL 或 G 再改变 RE 才能确保设置正确

6) 设定绘图 RAM 地址


格式	1	0	0	0	A3	A2	A1	A0
		A6	A5	A4	A3	A2	A1	A0

设定 GDRAM 地址到地址计数器 (AC), 先设置垂直位置再设置水平位置 (连续写入 2 字节数据来完成垂直与水平坐标的设置)。

垂直地址范围: AC6~AC0

水平地址范围: AC3~AC0

5. 初始化流程




第六章 屏幕与 DDRAM 地址的对应关系


	第 1 字	第 2 字	第 7 字	第 8 字
第一行	80H	81H	86H	87H
第二行	90H	91H	96H	97H
第三行	88H	89H	8EH	8FH
第四行	98H	99H	9EH	9FH

第七章 MCU 接口图


1. 并口直接访问方式



2. 并口间接访问方式



3. 串口接线图



第八章 示例程序

1. 并口直接访问方式

```

;*****
;CONTROLLER:ST7920(8BIT INTERFACE) *
;EDITOR:张炎 *
;MCU=89C52
;CRYSTAL:12MHz *
;RS=P2.1 RW=P2.0 E=/(WR*RD) *
;DB0-DB7=P0 *
;*****

DATWPEQU 0200H ;写数据通道
DATRPEQU 0300H ;读数据通道
COMWPEQU 0000H ;写数据通道
COMRPEQU 0100H ;读忙通道

COM EQU 30H ;控制字暂存单元
DAT EQU 31H ;显示数据暂存单元
CODER EQU 32H ;字符代码暂存单元
ADDR EQU 33H ;地址暂存单元

ORG 0000H
AJMP MAIN
ORG 0030H

MAIN:
MOV SP,#60H
LCALL DEL_20MS
LCALL DEL_20MS
LCALL INI ;进入初始化子程序

MOV ADDR,#80H
MOV DPTR,#WEL_1
LCALL W_LINE
MOV ADDR,#90H
MOV DPTR,#WEL_2
LCALL W_LINE
MOV ADDR,#88H
MOV DPTR,#WEL_3
LCALL W_LINE
MOV ADDR,#98H
MOV DPTR,#WEL_4

```

```
LCALL W_LINE
LCALL DEL_1500MS

MOV  ADDR,#80H
MOV  DPTR,#WEL_5
LCALL W_LINE
MOV  ADDR,#90H
MOV  DPTR,#WEL_6
LCALL W_LINE
MOV  ADDR,#88H
MOV  DPTR,#WEL_7
LCALL W_LINE
MOV  ADDR,#98H
MOV  DPTR,#WEL_8
LCALL W_LINE
LCALL DEL_1500MS

MOV  ADDR,#80H      ;
MOV  DPTR,#HZ_1     ;
LCALL W_LINE
MOV  ADDR,#90H      ;
MOV  DPTR,#HZ_2     ;
LCALL W_LINE
MOV  ADDR,#88H      ;
MOV  DPTR,#HZ_3     ;
LCALL W_LINE
MOV  ADDR,#98H      ;
MOV  DPTR,#HZ_4     ;
LCALL W_LINE
LCALL DEL_1500MS   ;延时后显示下一页

MOV  ADDR,#80H      ;
MOV  DPTR,#HZ_5     ;
LCALL W_LINE
MOV  ADDR,#90H      ;
MOV  DPTR,#HZ_6     ;
LCALL W_LINE
MOV  ADDR,#88H      ;
MOV  DPTR,#HZ_7     ;
LCALL W_LINE
MOV  ADDR,#98H      ;
MOV  DPTR,#HZ_8     ;
LCALL W_LINE
LCALL DEL_1500MS
```

```
LJMP MAIN
```

```
INI:
```

```
MOV COM,#30H ;功能设定,基本指令
LCALL WCOM
MOV COM,#30H ;基本指令,8-bit 模式,基本指令
LCALL WCOM
MOV COM,#0CH ;显示开,游标关,反白关
LCALL WCOM
MOV COM,#01H ;清除显示
LCALL WCOM
MOV COM,#06H ;进入设定点,游标 7 右移,画面不移动
LCALL WCOM
RET
```

```
W_LINE:
```

```
MOV COM,ADDR ;
LCALL WCOM ;
MOV R2,#16 ;16 个循环,连续写入 8 个汉字或是 16 个西文字符
```

```
W_L1:
```

```
MOV A,#00H
MOVC A,@A+DPTR
MOV CODER,A
LCALL WCODE
INC DPTR
DJNZ R2,W_L1
RET
```

```
WCOM:
```

```
PUSH DPH
PUSH DPL
MOV DPTR,#COMRP
```

```
WCOM1:
```

```
MOVX A,@DPTR
JB ACC.7,WCOM1
MOV DPTR,#COMWP
MOV A,COM
MOVX @DPTR,A
POP DPL
POP DPH
RET
```

```
WCODE:
```

```
PUSH DPH
PUSH DPL
MOV DPTR,#COMRP
WCODE1:
MOVX A,@DPTR
JB ACC.7,WCODE1
MOV DPTR,#DATWP
MOV A,CODER
MOVX @DPTR,A
POP DPL
POP DPH
RET
```

```
DEL_20MS:
MOV R0,#20
```

```
D2:
MOV R1,#200
```

```
D1:
NOP
NOP
NOP
DJNZ R1,D1
DJNZ R0,D2
RET
```

```
DEL_100MS:
LCALL DEL_20MS
LCALL DEL_20MS
LCALL DEL_20MS
LCALL DEL_20MS
LCALL DEL_20MS
RET
```

```
DEL_500MS:
LCALL DEL_100MS
LCALL DEL_100MS
LCALL DEL_100MS
LCALL DEL_100MS
LCALL DEL_100MS
RET
```

```
DEL_1500MS:
LCALL DEL_500MS
LCALL DEL_500MS
```

```

    LCALL DEL_500MS
    RET

```

```

WEL_1:
    DB "欢迎使用汉升液晶"

```

```

WEL_2:
    DB "WWW.HSLCM.COM.CN"

```

```

WEL_3:
    DB "86-0755-86114312"

```

```

WEL_4:
    DB "128X64 全点阵显示"

```

```

WEL_5:
    DB "图形以及字符显示"

```

```

WEL_6:
    DB "可显示 8X4 个中文"

```

```

WEL_7:
    DB "内带标准中文字库"

```

```

WEL_8:
    DB "含 8192 个中文字型"

```

```

HZ_1:
    DB "锦瑟无端五十弦，"

```

```

HZ_2:
    DB "一弦一柱思华年。"

```

```

HZ_3:
    DB "庄生晓梦迷蝴蝶，"

```

```

HZ_4:
    DB "望帝春心托杜鹃。"

```

```

HZ_5:
    DB "沧海月明珠有泪，"

```

```

HZ_6:
    DB "蓝田日暖玉生烟。"

```

```

HZ_7:
    DB "此情可待成追忆，"

```

```

HZ_8:
    DB "只是当时已惘然。"

```

```

END

```

2. 并口间接访问方式

```

;*****
;CONTROLLER:ST7920(8BIT INTERFACE)
;128×64 点阵

```

```

;EDITOR:张炎
;CRYSTAL:12MHz
;RS=P3.0 RW=P3.1 E=P3.2 D0-D7=P1
;*****
RS EQU P3.0 ;H=DATA,L=COM
RW EQU P3.1 ;H=READ,L=WRITE
E EQU P3.2 ;

COM EQU 30H ;控制字暂存单元
DAT EQU 31H ;显示数据暂存单元
CODER EQU 32H ;字符代码暂存单元
ADDR EQU 33H ;地址暂存单元

ORG 0000H
AJMP MAIN
ORG 0030H

```

MAIN:

```

MOV SP,#60H
MOV P3,#0F8H
LCALL DEL_20MS
LCALL DEL_20MS
LCALL INI ;进入初始化子程序

MOV ADDR,#80H
MOV DPTR,#WEL_1
LCALL W_LINE
MOV ADDR,#90H
MOV DPTR,#WEL_2
LCALL W_LINE
MOV ADDR,#88H
MOV DPTR,#WEL_3
LCALL W_LINE
MOV ADDR,#98H
MOV DPTR,#WEL_4
LCALL W_LINE
LCALL DEL_1500MS

MOV ADDR,#80H
MOV DPTR,#WEL_5
LCALL W_LINE
MOV ADDR,#90H
MOV DPTR,#WEL_6

```



```

LCALL W_LINE
MOV  ADDR,#88H
MOV  DPTR,#WEL_7
LCALL W_LINE
MOV  ADDR,#98H
MOV  DPTR,#WEL_8
LCALL W_LINE
LCALL DEL_1500MS

MOV  ADDR,#80H  ;
MOV  DPTR,#HZ_1 ;
LCALL W_LINE
MOV  ADDR,#90H  ;
MOV  DPTR,#HZ_2 ;
LCALL W_LINE
MOV  ADDR,#88H  ;
MOV  DPTR,#HZ_3 ;
LCALL W_LINE
MOV  ADDR,#98H  ;
MOV  DPTR,#HZ_4 ;
LCALL W_LINE
LCALL DEL_1500MS ;延时后显示下一页

MOV  ADDR,#80H  ;
MOV  DPTR,#HZ_5 ;
LCALL W_LINE
MOV  ADDR,#90H  ;
MOV  DPTR,#HZ_6 ;
LCALL W_LINE
MOV  ADDR,#88H  ;
MOV  DPTR,#HZ_7 ;
LCALL W_LINE
MOV  ADDR,#98H  ;
MOV  DPTR,#HZ_8 ;
LCALL W_LINE
LCALL DEL_1500MS

LJMP MAIN

```

INI:

```

MOV  COM,#30H ;功能设定,基本指令
LCALL WCOM
MOV  COM,#30H ;基本指令,8-bit 模式,基本指令
LCALL WCOM

```

```

MOV    COM,#0CH      ;显示开,游标关,反白关
LCALL WCOM
MOV    COM,#01H      ;清除显示
LCALL WCOM
MOV    COM,#06H      ;进入设定点,游标 7 右移,画面不移动
LCALL WCOM
RET

W_LINE:                ;写入一行 8 汉字子程序
MOV    COM,ADDR      ;
LCALL WCOM            ;
MOV    R2,#16         ;16 个循环,连续写入 8 个汉字或是 16 个西文字符

W_L1:
MOV    A,#00H
MOVC  A,@A+DPTR
MOV    CODER,A
LCALL WCODE
INC   DPTR
DJNZ  R2,W_L1
RET

WCOM:                    ;写指令子程序
CLR   RS
SETB  RW

WC1:
MOV   P1,#0FFH
SETB  E
MOV   A,P1
CLR   E
JB    ACC.7,WC1      ;查忙
CLR   RW
MOV   P1,COM
SETB  E
CLR   E
RET

WCODE:                    ;写代码子程序
CLR   RS
SETB  RW

WCO1:
MOV   P1,#0FFH
SETB  E
MOV   A,P1
CLR   E

```

```
JB ACC.7,WCO1
SETB RS
CLR RW
MOV P1,CODER
SETB E
CLR E
RET
```

DEL_20MS:

```
MOV R0,#20
```

D2:

```
MOV R1,#200
```

D1:

```
NOP
```

```
NOP
```

```
NOP
```

```
DJNZ R1,D1
```

```
DJNZ R0,D2
```

```
RET
```

DEL_100MS:

```
LCALL DEL_20MS
```

```
LCALL DEL_20MS
```

```
LCALL DEL_20MS
```

```
LCALL DEL_20MS
```

```
LCALL DEL_20MS
```

```
RET
```

DEL_500MS:

```
LCALL DEL_100MS
```

```
LCALL DEL_100MS
```

```
LCALL DEL_100MS
```

```
LCALL DEL_100MS
```

```
LCALL DEL_100MS
```

```
RET
```

DEL_1500MS:

```
LCALL DEL_500MS
```

```
LCALL DEL_500MS
```

```
LCALL DEL_500MS
```

```
RET
```

WEL_1:

```
DB "欢迎使用汉升液晶"
```

WEL_2:
DB "WWW.HSLCM.COM.CN"
WEL_3:
DB "86-0755-86114312"
WEL_4:
DB "128X64 全点阵显示"
WEL_5:
DB "图形以及字符显示"
WEL_6:
DB "可显示 8X4 个中文"
WEL_7:
DB "内带标准中文字库"
WEL_8:
DB "含 8192 个中文字型"

HZ_1:
DB "昨夜星辰昨夜风，"
HZ_2:
DB "画楼西畔桂堂东。"
HZ_3:
DB "身无彩凤双飞翼，"
HZ_4:
DB "心有灵犀一点通。"
HZ_5:
DB "隔座送钩春酒暖，"
HZ_6:
DB "分曹射覆蜡灯红。"
HZ_7:
DB "嗟余听鼓应官去，"
HZ_8:
DB "走马兰台类转蓬。"

END

3. 4 位并口模式

请参照 8 位并口间接访问方式以及 4 位并口时序图。不同之处只是将每个字节分两次送入，第一次送入高四位，第二次送入低四位。

4. 串口程序

PSB 接低时,串口模式被选择。在该模式下，只用两根线(SID 与 SCLK)来完成数据传输。当同时使用多颗 ST7920 时，CS 线被配合使用，CS 是高有效。

ST7920 的他不时钟 SCLK 有独立的操作时序，当多个连续的指令需要被送入时，指令执行时间需要被考虑。必须等待上一个指令执行完毕才送入下一个指

令，因为 ST7920 内部没有传送/接收缓冲区。

一个完整的串行传输周期由一下部分组成：

首先送入启动字节，送入 5 个连续的“1”用来启动一个周期，此时传输计数被重置，并且串行传输被同步。紧接的两个位指定传输方向（RW，确定读还是写）和传输性质（RS，确定是命令寄存器还是数据寄存器），最后的第八位是一个“0”。

送完启动字节之后，可以送入指令或是显示数据（或是字型代码）。指令或者代码是以字节为单位的，每个字节的内容（指令或数据）在被送入时分为两个字节来处理：高四位放在第一个字节的高四位，低四位放在第二个字节的高四位。无关位都补“0”。请参照第四章串行通讯时序图。

示例程序如下：

```

;*****
;CONTROLLER:ST7920(串行接口)
;128×64 点阵
;EDITOR: 张炎
;CRYSTAL:12MHz
;CS=P3.0 SID=P3.1 SCLK=P3.2
;*****
      CS   EQU  P3.0   ;H=DATA,L=COM
      SID  EQU  P3.1   ;H=READ,L=WRITE
      SCLK EQU  P3.2   ;

      COM  EQU  30H   ;控制字暂存单元
      DAT  EQU  31H   ;显示数据暂存单元
      CODER EQU  32H  ;字符代码暂存单元
      ADDR EQU  33H   ;地址暂存单元

      ORG  0000H
      AJMP MAIN
      ORG  0030H

MAIN:
      MOV  SP,#60H
      LCALL DEL_40MS
      LCALL INI

      MOV  ADDR,#80H
      MOV  DPTR,#WEL_1
      LCALL W_LINE
      MOV  ADDR,#90H
      MOV  DPTR,#WEL_2
      LCALL W_LINE
      MOV  ADDR,#88H

```

```
MOV DPTR,#WEL_3
LCALL W_LINE
MOV ADDR,#98H
MOV DPTR,#WEL_4
LCALL W_LINE
LCALL DEL_1500MS

MOV ADDR,#80H
MOV DPTR,#WEL_5
LCALL W_LINE
MOV ADDR,#90H
MOV DPTR,#WEL_6
LCALL W_LINE
MOV ADDR,#88H
MOV DPTR,#WEL_7
LCALL W_LINE
MOV ADDR,#98H
MOV DPTR,#WEL_8
LCALL W_LINE
LCALL DEL_1500MS

MOV ADDR,#80H ;
MOV DPTR,#HZ_1 ;
LCALL W_LINE
MOV ADDR,#90H ;
MOV DPTR,#HZ_2 ;
LCALL W_LINE
MOV ADDR,#88H ;
MOV DPTR,#HZ_3 ;
LCALL W_LINE
MOV ADDR,#98H ;
MOV DPTR,#HZ_4 ;
LCALL W_LINE
LCALL DEL_1500MS ;延时后显示下一页

MOV ADDR,#80H ;
MOV DPTR,#HZ_5 ;
LCALL W_LINE
MOV ADDR,#90H ;
MOV DPTR,#HZ_6 ;
LCALL W_LINE
MOV ADDR,#88H ;
MOV DPTR,#HZ_7 ;
LCALL W_LINE
```

```

MOV  ADDR,#98H      ;
MOV  DPTR,#HZ_8    ;
LCALL W_LINE
LCALL DEL_1500MS

```

```
LJMP MAIN
```

INI:

```

MOV  COM,#30H      ;功能设定,基本指令
LCALL WCOM
MOV  COM,#30H      ;基本指令,8-bit 模式,基本指令
LCALL WCOM
MOV  COM,#0CH      ;显示开,游标关,反白关
LCALL WCOM
MOV  COM,#01H      ;清除显示
LCALL WCOM
MOV  COM,#06H      ;进入设定点,游标 7 右移,画面不移动
LCALL WCOM
RET

```

W_LINE:

```

MOV  COM,ADDR
LCALL WCOM
MOV  R4,#16        ;连续写入 8 个中文或者 16 个西文字符

```

W_L1:

```

MOV  A,#00H
MOVC A,@A+DPTR
MOV  CODER,A
LCALL WCODE
INC  DPTR
DJNZ R4,W_L1
RET

```

WCOM:

```

LCALL STWC
MOV  A,COM
LCALL W4_D        ;送入高四位指令
LCALL W4_0        ;连续送入四个 0
LCALL W4_D        ;送入高四位指令
LCALL W4_0        ;连续送入四个 0
CLR  CS
LCALL DEL_2MS
RET

```

WCODE:

```
LCALL STWD
MOV  A,CODER
LCALL W4_D
LCALL W4_0
LCALL W4_D
LCALL W4_0
CLR  CS
LCALL DEL_2MS
RET
```

STWC:

```
SETB CS
SETB SID
MOV  R3,#5      ;连续送入 5 个"1",起始
```

STWC1:

```
SETB SCLK
CLR  SCLK
DJNZ R3,STWC1
CLR  SID
MOV  R3,#3
```

STWC2:

```
SETB SCLK      ;RW=0,RS=0,第八位"0"
CLR  SCLK
DJNZ R3,STWC2
RET
```

STWD:

```
SETB CS
SETB SID
MOV  R3,#5      ;连续送入 5 个"1",起始
```

STWD1:

```
SETB SCLK
CLR  SCLK
DJNZ R3,STWD1
CLR  SID      ;RW=0
SETB SCLK
CLR  SCLK
SETB SID      ;RS=1
SETB SCLK
CLR  SCLK
CLR  SID      ;第八位"0"
SETB SCLK
CLR  SCLK
```


RET

W4_D:

MOV R3,#4

W4_D1:

RLC A

MOV SID,C

SETB SCLK

CLR SCLK

DJNZ R3,W4_D1

RET

W4_0:

MOV R3,#4

W4_01:

CLR SID

SETB SCLK

CLR SCLK

DJNZ R3,W4_01

RET

DEL_2MS:

MOV R0,#2

D1:

MOV R1,#200

D2:

NOP

NOP

NOP

DJNZ R1,D2

DJNZ R0,D1

RET

DEL_40MS:

MOV R2,#20

D3:

LCALL DEL_2MS

DJNZ R2,D3

RET

DEL_200MS:

MOV R2,#100

D4:

LCALL DEL_2MS

```
DJNZ R2,D4
RET
```

DEL_500MS:

```
MOV R2,#250
```

D5:

```
LCALL DEL_2MS
DJNZ R2,D5
RET
```

DEL_1500MS:

```
LCALL DEL_500MS
LCALL DEL_500MS
LCALL DEL_500MS
RET
```

WEL_1:

```
DB "WWW.HSLCM.COM.CN"
```

WEL_2:

```
DB "欢迎使用汉升液晶"
```

WEL_3:

```
DB "086-755-86114312"
```

WEL_4:

```
DB "支持图形和文本显"
```

WEL_5:

```
DB "示, 可显示 8X4 行"
```

WEL_6:

```
DB "共 32 个中文, 内带"
```

WEL_7:

```
DB "标准中文字库, 内"
```

WEL_8:

```
DB "含 8192 个中文字型"
```

HZ_1:

```
DB " 桃      夭  "
```

HZ_2:

```
DB "诗经·国风·周南"
```

HZ_3:

```
DB "桃之夭夭灼灼其华"
```

HZ_4:

```
DB "之子于归宜其室家"
```

HZ_5:

```
DB "桃之夭夭有蕢其实"
```

HZ_6:

DB "之子于归宜其家室"
 HZ_7:
 DB "桃之夭夭其叶蓁蓁"
 HZ_8:
 DB "之子于归宜其家人"

 END

液晶使用注意事项

附录 1：各种背光电参数

序号	类型	工作电压	每颗灯电流	备注
1	LED 黄绿侧光	4.1V	10mA	
2	LED 黄绿底光	4.1V	10mA	
3	LED 白色侧光	3.0V	15mA	
4	EL 黄光	70VAC	-	需逆变器
5	EL 蓝光	70VAC	-	需逆变器
6	CCFL	逆变器 5V	-	需逆变器

注：以上参数仅作参考，不同型号的背光会稍有差异，具体见相关产品资料。

附录 2：注意事项

十分感谢您购买汉升公司的产品，在使用前请您首先仔细阅读以下注意事项，以免给您造成不必要的损失，您在使用过程中遇到困难时，请拨打我们的服务电话 0755-86114312-8072，我们将尽力为您提供服务和帮助。

1. 处理保护膜

在装好的模块成品表面贴有一层保护膜，以防在装配时沾污显示表面，在整机装配结束前不得撕去，以免弄脏或损坏表面。

2. 加装衬垫

在模块和前面板之间最好加装一块约 0.1 毫米左右的衬垫。面板还应保持平整，以免在装配后产生扭曲，并可提高其抗振性能。

3. 严防静电

模块中的控制、驱动电压是很低、微功耗的 CMOS 电路，极易被静电击穿，静电击穿是一种不可修复的损坏，而人体有时会产生高达几十伏或上百伏的静电，所以，在操作、装配以及使用中都应极其小心，严防静电。为此：

- (1) 不要用手随意去摸外引线、电路板上的电路及金属框。
- (2) 如必须直接接触时，应使人体与模块保持在同一电位，或使人体良好接地。
- (3) 焊接使用的烙铁及装配使用的电动工具必须良好接地，没有漏电。
- (4) 不得使用真空吸尘器进行清洁处理，因为它会产生很强的静电。
- (5) 空气干燥也会产生静电，因此，工作间湿度应在 RH60%以上。
- (6) 取出或放回包装袋或移动位置时，也需小心，防止产生静电。不要随意更换包装或舍弃原包装。

4. 装配操作时的注意事项

- (1) 模块是经过精心设计组装而成的，请勿随意自行加工、修整。
- (2) 金属框爪不得随意扭动、拆卸。
- (3) 不要随意修改加工 PCB 板外形、装配孔、线路及其部件。
- (4) 不得修改导电胶条。
- (5) 不得修改任何内部支架。
- (6) 不要碰、摔、折曲、扭动模块。

5. 焊接

在焊接外引线时，应按如下规程进行操作。

- (1) 烙铁头温度小于 280 度。
- (2) 焊接时间不超过 4 秒。
- (3) 焊接材料：共晶型、低熔点。
- (4) 不要使用酸性助焊剂。
- (5) 重复焊接不要超过三次，且每次重复需间隔 5 分钟。

6. 模块的使用与保养

- (1) 模块的外引线决不允许接错，在您想调试液晶模块时，请注意正确接线，尤其

是正负电源的接线不能接错, 否则可能造成过流、过压烧电路上的芯片等对液晶模块元器件有损的现象。

- (2) 模块在使用时, 接入电源及断开电源, 必须在正电源稳定接入以后才能输入信号电平。如在电源稳定前或断开后输入信号电平, 有可能损坏模块中的 IC 及电路。
- (3) 点阵液晶模块显示时的对比度、视角与温度、驱动电压的关系很大, 所以, 如果驱动电压过高, 不仅会影响显示效果, 还会缩短模块的使用寿命。
- (4) 因为液晶材料的物理特性, 液晶的对比度会随温度的变化而相应变化, 所以, 您加的负压也应随温度作相应调整。大致是温度变化 10 度, 电压变化 1 伏。为满足这一要求, 您可以做一个温度补偿电路, 或者安排一个电位器, 随温度调整负电压值。
- (5) 不应在规定工作温度范围外使用, 并且不应在超过存储极限温度的范围外存储。如果温度低于结晶温度, 液晶就会结晶, 如果温度过高, 液晶将变成各向同性的液晶, 破坏分子取向, 使器件报废。
- (6) 用力按显示部分, 会产生异常显示。这时切断电源, 稍待片刻重新上电, 即恢复正常。
- (7) 液晶显示器件或模块表面结雾时, 不要通电工作, 因为这将引起电极化学反应, 产生断线。
- (8) 长期用于阳光及强光下时, 被遮部分会产生残留现象。

7. 液晶模块的存储

若长期 (如几年以上) 存储, 我们推荐以下方式:

- (1) 装入聚乙稀口袋 (最好有防静电涂层) 并将口封住
- (2) 在 -10°C --- $+35^{\circ}\text{C}$ 之间存储。
- (3) 放在暗处, 避强光。
- (4) 决不能在表面压放任何物品。
- (5) 严格避免在极限温度/湿度条件下存放。

8. 有限责任和保修

如果汉升公司和客户没有发生任何协议, 汉升公司将从发货日期算起一年内依据汉升公司液晶显示模块接受标准 (按要求提供复印件) 更换或修理功能性故障的液晶显示模块。

外观/视觉毛病必须从发货日计起 90 天内送返汉升公司。日期的确认将根据货运文件。汉升公司保证的责任限于上述提及项目的维修和更换, 汉升公司不对突发性事件负责任。

保修是以上述注意事项未被忽视为先决条件的, 典型的违反例子如下:

- (1) 断裂的液晶显示屏玻璃。
- (2) 线路板孔修改或损坏。
- (3) 线路板布线损坏。
- (4) 电路修改, 包括元件的增加。

(5) 线路板随意研磨、雕刻或油漆。

(6) 焊接或更改玻璃框。

模块维修将基于双方协议下列出给顾客的清单。模块必须与防静电包装和故障详细陈述一起送回。顾客安装的连接器或电缆必须坏线路板孔，线路和引线端条件下全部移去在不破坏线路板孔，线路和引线端条件下全部移去。